
 Agenda
 
 | tstats summariesonly=t

 values(conf.section)

 from datamodel=conf

 where conf.talk=TRU1133

 by conf.section_number

1) Quick Reintroduction to tstats
 allow_old_summaries = true

2) Cool Things tstats Can Do
 TERM(cool=1)

3) Converting Searches to tstats
 | convert

4) “Pivot” Job Inspector
 | pivot

5) Optional Arguments
 append = true

(Re-)Intro to 
| tstats

Datamodel-based tstats:

| tstats count
from datamodel=Web
by host

- some flexibility at the 
expense of prep & 
rebuild efforts

- very fast with DMA

Index-based tstats:

| tstats count 
where index=web 
by host

- limited to index-time
- very fast

Three Types of Stats-y Searches*

Event searches:

index=web
| stats count by host

- infinite flexibility
- speed penalties

*There’s also mstats for metrics indexes, that’s a topic for another talk...

Filtering

• Filter by indexed fields
 | tstats count where index=_internal source=*metrics.log*

• Filter by simple indexed terms
 | tstats count where index=_internal source=*metrics.log* typingqueue

• Filter by complex indexed terms
 | tstats count where index=_internal source=*metrics.log* "blocked=true"

•😥 
 | tstats count where index=_internal source=*metrics.log* TERM(blocked=true)

•😅 TERM() forces Splunk to treat blocked=true as one indexed term, ignoring the minor segmenter =

PREFIX() Fun

• This does not work, there is no indexed field name:
 | tstats count where index=_internal source=*metrics.log* TERM(blocked=true) by name

• PREFIX(name=) will ad-hoc-interpret the indexed term name=typingqueue as a key-value pair (v8.0+)
 | tstats count where index=_internal source=*metrics.log* TERM(blocked=true) by PREFIX(name=)

•🤯

PREFIX() Magic

• “Traditional” search to chart queue sizes:
 index=_internal source=*metrics.log* TERM(group=queue) 
| timechart span=15m max(current_size_kb) by name

 Observed 80k Events/s per indexer in a real-world environment… not bad, but can tstats help?

• PREFIX() works for aggregations too!
 | tstats max(PREFIX(current_size_kb=)) as max_size_kb 
  where index=_internal source=*metrics.log* TERM(group=queue) 
  by _time span=15m PREFIX(name=)
| timechart span=15m max(max_size_kb) by "name="

 Same environment returned 2-3M EPS per indexer, 30x speedup

 
• But remember, your raw event layout matters: name=typingqueue works while name="typingqueue" 

breaks PREFIX() due to segmentation

Basic Conversion

 index=main
| stats count by sourcetype

 

| tstats count where index=main by 
sourcetype

Event Search tstats Search

Slightly More Complex Conversion

 index=pan_logs
| stats count

| tstats count where index=pan_logs 

Event Search tstats Search

Slightly More Complex Conversion

 index=pan_logs
| stats count by host

 

| tstats count where index=pan_logs 
by host

Event Search tstats Search

Slightly More Complex Conversion

 index=pan_logs
| timechart span=1d count by host

 

| tstats prestats=t count where 
index=pan_logs by _time span=1d host
| timechart span=1d count by host

Event Search tstats Search

Slightly More Complex Conversion

 index=pan_logs
| eval _time=_indextime
| timechart span=1d count by host

 

| tstats prestats=t count where 
index=pan_logs by _indextime host
| eval _time=_indextime
| timechart span=1d count by host

Event Search tstats Search

Slightly More Complex Conversion

 index=pan_logs
| eval _time=_indextime
| timechart span=1d count by host

 

| tstats prestats=t count where 
index=pan_logs by _indextime host
| eval _time=_indextime
| timechart span=1d count by host

Event Search tstats Search

70x 
Faster

Conversion Algorithm

1) What is the end result?

2) Are the fields needed available at index time?
• | walklex index="<index>" type=field

| search NOT field=" *"
| stats list(distinct_values) by field

3) Rearrange the search so that the aggregation commands are in the tstats command

4) Use prestats=true when wanting to run a timechart command

5) Add index, source, sourcetype, etc. filters into the WHERE clause

From search to tstats

Let’s Pivot

 Data Models
 Job Inspector to 
the Rescue

 Find the | tstats search from a 
working Pivot using the Job 
Inspector

 Data Models
 Job Inspector to 
the Rescue

 Note that this only works when 
the DM is accelerated - pivot 
on an unaccelerated DM is 
converted to regular search

Expanding Pivot

 Run pivot in search

 Mac: cmd + shift + E
Windows: ctrl + shift + E

 Pivot will expand into tstats or generic 
search, depending on DM acceleration

 

 | pivot pan_firewall log count(log) AS "Count 
of Firewall Logs" ROWSUMMARY 0 COLSUMMARY 0 
SHOWOTHER 1

 BECOMES

 | tstats prestats=true summariesonly=false 
allow_old_summaries=false count as "Count of 
Firewall Logs" FROM datamodel=pan_firewall 
WHERE nodename=log 
| stats dedup_splitvals=t count AS "Count of 
Firewall Logs" 
| fillnull "Count of Firewall Logs" 
| fields + "Count of Firewall Logs"

Explore Your 
Options

Datamodel-specific Options

 summariesonly=true
• Use accelerated data only, do not fill in the gaps for the last few minutes
• Great for dashboards displaying 24 hours of data
 | tstats summariesonly=true estdc(All_Traffic.src) from datamodel=Network_Traffic by All_Traffic.dvc

 allow_old_summaries=true
• Allows tstats to use existing older accelerated data even if the DM changed, e.g. an eval-field was refined
• Favours speed over using the latest search-time knowledge objects
• Most relevant in combination with summariesonly=true, avoids entirely empty results after a change
 | tstats summariesonly=true allow_old_summaries=true count from datamodel=Authentication by _time span=1h

Good Options for All

 fillnull_value="Buttercup"
• Tells tstats what do do when events don’t have values for a group-by-field
• Default behaviour: Ignore those events!
 | tstats fillnull_value="Buttercup" count where index=_internal by user

 span=1h
• Used when grouping by _time, almost like in the timechart and bin commands
• Aggregates slices of time together by the given span instead of automatically based on time range
 | tstats count from datamodel=Authentication by _time span=1h

• Note: Unlike timechart, tstats will not generate rows for empty spans

Map-Reduce Shenanigans

 prestats=true
• Allows you to specify the Map- and Reduce-parts of tstats independently from each other
• Enables lots of cool hacks coming up in a minute
 | tstats prestats=true count from datamodel=Web by Web.src | 󰩃 | stats count by Web.src

 append=true
• Lets you concatenate multiple tstats result sets on the indexers, prestats=true only
• Allows 7.x-compatible emulation of fillnull_value without append command:
 | tstats prestats=true count where index=_internal user=* by user
| tstats prestats=true append=true count where index=_internal NOT user=*
| fillnull value="Buttercup" user
| stats count by user

prestats=true Basics

• Every search has a Map- and a Reduce-phase
• Job inspector: phase0 & phase1
• Event search phase0 is “everything including the first stats”, phase1 is “everything from the first stats”
• tstats is its own generating command and stats combined, therefore tstats is split up automatically:
 search: | tstats count where index=_internal by sourcetype
 phase0: | tstats prestats=t <snip> count where index=_internal groupby sourcetype
 phase1: | tstats <snip> count WHERE index=_internal BY sourcetype

• That’s equivalent to specifying both phases manually:
 | tstats prestats=t count where index=_internal by sourcetype
| stats count by sourcetype

prestats=true Under the Hood

• What if you “forget” the manual phase1 / Reduce?



© 2021 SPLUNK INC.

UPDATED Splunk .conf21  |  TMPLT-FY22-102

prestats=true Under the Hood

• What if you “forget” the manual phase1 / Reduce?

• psrsvd_gc = “prestats-reserved global count”, a partial contribution to the total count for splunkd
• You can do SPL with those fields! On the Indexers!
• Touching group-by fields (sourcetype) is usually safe, don’t touch aggregations (count / psrsvd_gc)

prestats=true Example: Lookups

• Let the Searchhead do the lookup work all on its own
 | tstats sum(All_Traffic.bytes) as bytes from datamodel=Network_Traffic by All_Traffic.src_ip
| lookup dnslookup clientip as "All_Traffic.src_ip" OUTPUT clienthost
| search clienthost="*.splunk.com"
| stats sum(bytes) as splunk_bytes

• Let the Indexers share the lookup work between themselves in parallel
 | tstats prestats=t sum(All_Traffic.bytes) from datamodel=Network_Traffic by All_Traffic.src_ip
| lookup dnslookup clientip as "All_Traffic.src_ip" OUTPUT clienthost
| search clienthost="*.splunk.com"
| stats sum(All_Traffic.bytes) as splunk_bytes

• Note: Renames with prestats=t are ignored, the reduce-stats has to refer to the original name

Indexers
Searchhead

Indexers
Searchhead

prestats=true Example: Naïve Approach

 Enterprise Security Content Update: SQL Injection with Long URLs
• Load all URLs from the Web datamodel to the searchhead, apply complex filtering
 | tstats summariesonly=t count from datamodel=Web by Web.url
| eval num_sql_cmds = <complex eval to count SQL-y looking things>
| where num_sql_cmds > 3

 
 

• Over 1GB of data returned to SH, 2.5M rows need to be merged and sorted before eval
• Single SH doing filtering

prestats=true Example: Smart Approach

 Enterprise Security Content Update: SQL Injection with Long URLs
• Offload the complex filtering to the indexers, only load matching URLs to the searchhead
 | tstats summariesonly=t prestats=t count from datamodel=Web by Web.url
| eval num_sql_cmds = <complex eval to count SQL-y looking things>
| where num_sql_cmds > 3
| stats count by Web.url

 
 

• Under 10MB of data returned to SH
• Parallel filtering on every indexer
• 6x speedup

Key Takeaways

What to Remember

• tstats works on indexed fields and datamodels

• ctrl+shift+E or cmd+shift+E will expand the pivot search and 
show tstats

• prestats=t is great when additional aggregations follow

• FROM specifies the datamodel

• WHERE filters events

• PREFIX() and TERM() are magic

• Be mindful of high cardinality intermediate results

 Helpful 
Links

• More on the Job Inspector: 
https://conf.splunk.com/watch/conf-online.html?searc
h=TRU1143C 

• tstats Reference Docs: 
https://docs.splunk.com/Documentation/Splunk/latest/
SearchReference/Tstats 

• pivot Reference Docs: 
https://docs.splunk.com/Documentation/Splunk/latest/
SearchReference/Pivot 

• Additional Background on Map-Reduce: 
https://en.wikipedia.org/wiki/MapReduce

https://conf.splunk.com/watch/conf-online.html?search=TRU1143C
https://conf.splunk.com/watch/conf-online.html?search=TRU1143C
https://docs.splunk.com/Documentation/Splunk/latest/SearchReference/Tstats
https://docs.splunk.com/Documentation/Splunk/latest/SearchReference/Tstats
https://docs.splunk.com/Documentation/Splunk/latest/SearchReference/Pivot
https://docs.splunk.com/Documentation/Splunk/latest/SearchReference/Pivot
https://en.wikipedia.org/wiki/MapReduce

Thank You
SESSION SURVEY

Please provide feedback via the


